
Lab # 3


LAB # 3

GATE-LEVEL MODELING

OBJECTIVE

To describe digital circuits in Verilog using gate level modeling constructs.

THEORY

Verilog allows the designer of digital circuits four levels of abstraction namely, behavioral or algorithmic level, dataflow level, gate level and switch level.  Most digital design is now done at gate level or higher levels of abstraction.  At gate level, the circuit is described in terms of gates (e.g. and, nand, or etc.).  Hardware design at this level is intuitive for a user with basic knowledge of digital logic design because it is possible to see a one-to-one correspondence between the logic circuit diagram and the Verilog description.

In this lab we shall work with logic gate primitives provided in Verilog and learn how to construct a Verilog description from the logic diagram of the circuit.  We shall also learn how to model the delays in actual gates in Verilog.

Gate Types

A logic circuit can be designed by use of logic gates.  Verilog supports basic gates as predefined primitives.  These primitives are instantiated like modules except that they are predefined in Verilog and do not need a module definition.  All logic circuits can be designed by using gates.  There are two classes of basic gates: and/or gates and buf/not gates.

And/Or Gates

And/Or gates have one scalar (i.e. 1-bit) output and multiple scalar inputs.  The first terminal in the list of gate terminals is an output and the other terminals are inputs.  The output of the gate is evaluated as soon as one of the input changes.  The and/or gates available in Verilog are shown below.

and
or
nand


nor
xor
xnor
We consider gates with two inputs.  The output terminal is denoted by out.  Input terminals are denoted by i1 and i2.

These gates are instantiated to build logic circuits in Verilog.  Examples of gate instantiations are shown below.  In the example, for all instances, OUT is connected to the output OUT, and IN1 and IN2 are connected to the two inputs i1 and i2 of the gate primitives.  Note that the instance names do not need to be specified for primitives.  This lets the designer instantiates hundreds of gates without giving them a name.

More than two input can be specified in a gate instantiation.  Gates with more than two inputs are instantiated by simply adding more input ports in the gate instantiation.  Verilog automatically instantiates the appropriate gate.

wire OUT, IN1, IN2;

// basic gate instantiations

and a1(OUT, IN1, IN2);

nand na1(OUT, IN1, IN2);

or or1(OUT, IN1, IN2);

nor nor1(OUT, IN1, IN2);

xor x1(OUT, IN1, IN2);

xnor nx1(OUT, IN1, IN2);

// More than two inputs; 3 input nand gate

nand na1_3inp(OUT, IN1, IN2, IN3);

// gate instantiation without instance name

and (OUT, IN1, IN2); // legal gate instantiation

Buf/Not Gates

Buff/not gates have one scalar input and one or more scalar outputs. The last terminal in the port list is connected to the input. Other terminals are connected to the outputs. We will discuss gates that have one input and one output.

Two basic buff/not gate primitives are provided in Verilog.

buf
not 

These gates are instantiated in Verilog as shown below. Notice that these gates can have multiple outputs but exactly one input, which is the last terminal in the port list.

// basic gate instantiation

buf b1(OUT1, IN);

not n1(OUT1, IN);

// More than two outputs

buf b1_2out(OUT1, OUT2, IN);

// gate instantiation without instance name

nor (OUT1, IN);
// legal gate instantiation

Example

Having understood the various types of gates available in Verilog, we will discuss a real example that illustrates design of gate-level digital circuits. 

Gate-level multiplexer

We will design a 4-to-1 multiplexer with 2 select signals. Multiplexers serve a useful purpose in logic design. They can connect two or more sources to a single destination. They can also be used to inplement Boolean functions. We will assume for this example that signals s1 and s0 do not get the value x or z. The I/O diagram and the truth table for the multiplexer are shown in Figure 3.1. The I/O diagram will be useful in setting up the port list for the multiplexer.

	s1
	s2
	Out

	0
	0
	i0

	0
	1
	i1

	1
	0
	i2

Fig 3-1: 4-to-1 Multiplexer

We will implement the logic for the multiplexer using basic logic gates. The logic diagram for the multiplexer is shown in Figure 3.2.

Fig 3.2: Logic Diagram for Multiplexer

The logic diagram has a one-to-one correspondence with the Verilog description. The Verilog description for the multiplexer is shown below. Two intermediate nets, s0n and s1n, are created; they are complements of input signals s1 and s0.  Internal nets y0, y1, y2, y3 are also required. Note that instance names are not specified for primitive gates, not, and, and or.  Instance names are optional for Verilog primitives but are mandatory for instance of user-defined modules.

// Module 4-to-1 multiplexer.  Port list is taken exactly from the I/O diagram

module mux4_to_1 (out, i0, i1, i2, i3, s1, s0);

// Port declarations from the I/O diagram

output out;

input i0, i1, i2, i3;

input s0, s1;

// Internal wire declarations

wire s1n, s0n;

wire y0, y1, y2, y3;

// Gate instantiations

// Create s1n and s0n signals

not (s1n, s1);

not (s0n, s0);

// 3-input and gates instantiated

and (y0, i0, s1n, s0n);

and (y1, i1, s1n, s0);

and (y2, i2, s1, s0n);

and (y3, i3, s1, s0);

// 4-input or gate instantiated

or (out, y0, y1, y2, y3);

endmodule

This multiplexer can be tested with the stimulus shown below. The stimulus checks that each combination of select signals connects the appropriate input to the outputs. The signal OUTPUT is displayed one time unit after it changes.  System task $monitor could also be used to display the signals when they change values.

// Define the stimulus module (no ports)

module stimulus;

// Declare variables to be connected to inputs

reg IN0, IN1, IN2, IN3;

reg S1, S0;

// Declare output wire

wire OUTPUT;

// Instantiate the multiplexer

mux4_to_1 mymux(OUTPUT, IN0, IN1, IN2, IN3, S1, S0);

// Stimulate the inputs

initial

begin

  // set input lines

  IN0 = 1; IN1 = 0; IN2 = 1, IN3 = 0;

  #1 $display(“IN0 = %b, IN1 = %b, IN2 = %b, IN3 = %b\n”, IN0, IN1, IN2, IN3);

  // choose IN0

  S1 = 0; S0 = 0;

  #1 $display(“S1 = %b, S0 = %b, OUTPUT = %b \n”, S1, S0, OUTPUT);

  // choose IN1

  S1 = 0; S0 = 1;

  #1 $display(“S1 = %b, S0 = %b, OUTPUT = %b \n”, S1, S0, OUTPUT);

  // choose IN2

  S1 = 1; S0 = 0;

  #1 $display(“S1 = %b, S0 = %b, OUTPUT = %b \n”, S1, S0, OUTPUT);

  // choose IN3

  S1 = 1; S0 = 1;

  #1 $display(“S1 = %b, S0 = %b, OUTPUT = %b \n”, S1, S0, OUTPUT);

end

endmodule

The output of the simulation is shown below. Each combination of the select signals is tested.

IN0 = 1, IN1 = 0, IN2 = 1, IN3 = 0

S1 = 0, S0 = 0, OUTPUT = 1

S1 = 0, S0 = 1, OUTPUT = 0

S1 = 1, S0 = 0, OUTPUT = 1

S1 = 1, S0 = 1, OUTPUT = 0

EXERCISE

1. A 2-input xor gate can be built from and, or and not gates.  Construct an xor gate module in Verilog that realizes the logic function z = xy’ + x’y.  Inputs are x and y, and z is the output.  Write a stimulus module that exercises all four combinations of x and y inputs.

2. The 1-bit full adder can be expressed in a sum of products form.

sum = a.b.c_in + a’.b.c_in’ + a’.b’.c_in+a.b’.c_in’

c_out =  a.b + b.c_in + a.c_in


Write the Verilog description for the circuit.

